Ein Beispiel zur Integration von Funktionen mehrerer Variablen und zur grafischen Darstellung von Körpern im Raum mit *Mathematica*

R. Brigola, Sommersemester 2013

Wir betrachten folgende Aufgabe:

Berechnen Sie das Volumen des Körpers, der entsteht, wenn man aus einem Kreiskegel der Höhe h mit der Grundfläche vom Radius R einen Zylinder mit Radius R/2 herausbohrt.
Die Zylinderachse sei parallel zur Kegelachse und besitze von ihr den Abstand R/2.
Es soll das Volumen des herausgebohrten Körpers berechnet werden. Man verwende dabei Zylinderkoordinaten.

Bevor die Lösung berechnet wird, wollen wir uns ansehen, wie man den genannten Körper zweckmäßig beschreibt, um die Integrationsgrenzen für das Volumenintegral festzustellen. Außerdem wollen wir Körper mit Hilfe von *Mathematica* grafisch darstellen.

Der Körper ist der Durchschnitt der beiden Mengen, die mit Zylinderkoordinaten wie folgt beschrieben werden können:

$M1=\{ (r,\phi,z) \mid 0\le r\le R,\ 0\le\phi<2\pi,\ h/R\ r\le z\le h \}$ = Kegel, dessen Spitze im Nullpunkt steht
und welcher rotationssymmetrisch zur z-Achse ist

$M2=\{ (r,\phi,z) \mid 0\le r\le R\ \mathrm{Sin}[\phi],\ 0\le\phi\le\pi,\ 0\le z\le h \}$ = Zylinder, wie oben beschrieben

====> $M1\cap M2 = \{ (r,\phi,z) \mid 0\le r\le R\ \mathrm{Sin}[\phi],\ 0\le\phi\le\pi,\ h/R\ r \le z\le h \}$

(Beim Durchschnitt müssen alle obigen Ungleichungen gemeinsam erfüllt sein, werden also logisch mit UND verknüpft)

Damit kann man nun sofort das Volumenintegral hinschreiben und berechnen. Wir machen es hier mit *Mathematica.* Integriert wird bzgl. Zylinderkoordinaten. Man beachte, dass bei *Mathematica* die Reihenfolge der Integration mit der am weitesten rechts stehenden Variablen beginnt usw. (Vgl. hierzu die Help-Seite zum Befehl Integrate).

```
Integrate[r, {ϕ, 0, π}, {r, 0, R Sin[ϕ]}, {z, h / R r, h}]
```

$$\frac{1}{36} h \, (-16 + 9\,\pi)\, R^2$$

Hier also das gesuchte Volumen
(Ein Test, ob das Ergebnis eine Volumeneinheit hat, empfiehlt sich als Prüfung):

Grafische Darstellung des Körpers mit *Mathematica*

Mathematica bietet den Befehl RegionPlot3D zur Darstellung von Körpern im Raum an.
Dieser Befehl (bzw. das dahinter stehende Programm) ist aber (bisher) auf die Beschreibung von Körpern in kartesischen Koordinaten beschränkt.

Um auch Körper darzustellen, die in Kugelkoordinaten oder Zylinderkoordinaten beschrieben sind, verwenden wir eine wie folgt programmierte Neudefinition des Befehls RegionPlot3D.

Das Beispiel zeigt schon ein wenig, wie man mit *Mathematica* auch komplexere Aufgaben “programmieren kann”.

Der neue Befehl heißt newRegionPlot3D und beinhaltet vor dem Aufruf des in *Mathematica* vorhandenen Befehls RegionPlot3D eine ggf. notwendige Koordinatentransformation von Kugel- oder Zylinderkoordinaten in kartesiche Koordinaten.

Die genaue Wirkung der Prozedur können an Programmierung interessierte Leser leicht mit den Help-Seiten zu den verwendeten Befehlen im Einzelnen erschließen. Der große Vorteil bei der Verwendung der schon vorhandenen Routine RegionPlot3D ist, dass man die von Mathematica erzeigte Grafik in ganz verschiedenen Perspektiven betrachten kann, ohne sich selbst um die Programmierung von Perspektivenveränderungen, Skalierungen etc. - ein ja durchaus kompliziertes Geschäft - noch kümmern zu müssen.

Referenz: Die nachfolgende Modifikation des RegionPlot3D-Befehls stammt nicht von mir, sondern von der URL:

http://mathematica.stackexchange.com/questions/16421/regionplot3d-in-cylindrical-or-spherical-coordinates/16435#16435

Bei Interesse an der neuen Technologie des ***3D-Druckens****: Mathematica unterstützt auch STL als Export-Format. Man beobachte einfach Weiterentwicklungen in diese Richtung. Ein Leselink ist (vorläufig) etwa die Referenz:* http://www.segerman.org/3d_printing_notes.html

```
newRegionPlot3D[expr_, {u_, umin_, umax_}, {v_, vmin_, vmax_},
  {w_, wmin_, wmax_}, tr_, opts___] := Module[{x, y, z, newExpr, xyz},
  newExpr = TransformedField[tr → "Cartesian", expr, {u, v, w} → {x, y, z}];
  xyz = CoordinateTransform[tr → "Cartesian", {u, v, w}];
  {xmax, ymax, zmax} = NMaxValue[
      {#, umin < u < umax && vmin < v < vmax && wmin < w < wmax}, {u, v, w}] & /@ xyz;
  {xmin, ymin, zmin} = NMinValue[{#, umin < u < umax &&
         vmin < v < vmax && wmin < w < wmax}, {u, v, w}] & /@ xyz;
  RegionPlot3D[newExpr, {x, xmin, xmax}, {y, ymin, ymax}, {z, zmin, zmax}, opts]]
```

Hier nun die Darstellung des Körpers, dessen Volumen wir schon kennen.
Im Beispiel sind R=h=2 gesetzt:

```
R = 2; h = 2;
```

```
plot1 = newRegionPlot3D[ h / R ro ≤ z && z ≤ h && 0 ≤ ro && ro ≤ R Sin[phi],
  {ro, 0, R}, {phi, 0, Pi}, {z, 0, h}, "Cylindrical",
  PlotPoints → 100, BoxRatios → {1, 1 / 2, 1}, ViewPoint → {-3, 1, 1}]
```



Wenn Sie die Grafik exportieren wollten, um sie etwa andernorts zu importieren und zu verwenden, können Sie dies mit dem Export-Befehl. erreichen. Nachfolgend als Beispiel der Export als JPEG-Grafik (für gute Grafik, die Sie einmal in Arbeiten einbetten wollen, empfehle ich EPS oder, wenn solche Files zu groß werden, dann TIFF als Export-Format).

SetDirectory[“drive:\your_directory\subdirectory”]   (* Legt Ihr Ausgabe-Verzeichnis fest *)
Export[“KoerperBeispiel.jpeg”,plot1,”EmbeddedFonts”→False] (* Export als JPEG*)

Weiteres Beispiel: Da wir nun wissen, wie man prinzipiell an Aufgaben des behandelten Typs herangehen kann, eine neue Aufgabe für Sie als Übung:

Stellen Sie noch eine Kugel vom Radius h/4 auf den Nullpunkt. Die Kugel schneidet wieder den schon betrachteten Körper. Nehmen Sie diesen Durchschnitt heraus und stellen Sie den entstehenden Körper wieder grafisch dar.

Berechnen Sie das Volumen dieses Restkörpers.

Wir betrachten 2 Bilder: 1) Den Durchschnitt der Kugel mit unserem Körper von oben

2) Den Körper ohne diesen Durchschnitt, dessen Volumen
Sie berechnen sollen:

Wenn Sie selbst dieses Notebook mit *Mathematica* evaluieren, können Sie die Grafik von allen Seiten ansehen.

Bild 1:

```
plot2 = newRegionPlot3D[
  h / R ro ≤ z && z ≤ h && 0 ≤ ro && ro ≤ R Sin[phi] && ro^2 + z^2 - h / 2 z ≤ 0,
  {ro, 0, h / 2}, {phi, 0, Pi}, {z, 0, h / 2}, "Cylindrical",
  PlotPoints → 100, BoxRatios → {1, 1 / 2, 1 / 2}, ViewPoint → {-9, 1, 3 / 4}]
```



Bild 2

```
plot3 = newRegionPlot3D[ h / R ro ≤ z && z ≤ h && 0 ≤ ro &&
   ro ≤ R Sin[phi] && ro^2 + z^2 - h / 2 z > 0, {ro, 0, R}, {phi, 0, Pi},
  {z, 0, h}, "Cylindrical", PlotPoints → 100, ViewPoint → {-9, -4, 2}]
```



Ich hoffe, Sie finden Gefallen an den Möglichkeiten, die *Mathematica* bietet und nutzen die Möglichkeit, wenn Sie *Mathematica* auf Rechnern Ihrer Hochschule zur Verfügung haben. Vielleicht versuchen Sie einmal, selbst ein paar Grafik-Prozeduren zu schreiben.

Als Vorschlag etwa: Schreiben Sie eine Modifikation der obigen Routine, so dass Sie Körper möglichst einfach darstellen können, von denen gewisse Teile in kartesischen Koordinaten, andere wiederum in Kugelkoordinaten und weitere evtl. in Zylinderkoordinaten am zweckmäßigsten beschrieben werden können. (Man denke etwa an ein aus Quadern, Zylinder- und Kugel-Teilen zusammengesetztes Gebilde.)

Nicht vergessen: Jetzt noch das Volumen des zuletzt dargestellten Körpers berechnen.